N70-32313

NASA TECHNICAL MEMORANDUM

NASA TM X-2039

NASA TM X-2039

THERMAL NEUTRON ABSORBER BUILDUP IN IRRADIATED BERYLLIUM

by Lawrence E. Peters, Jr.

Lewis Research Center
Cleveland, Ohio 44135

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • JULY 1970

1. Report No. NASA TM X-2039	2. Government Accession No.	3. Recipient's Catalog No.
4. Title and Subtitle THERMAL NEUTRON ABSORBER BUILDUP IN IRRADIATED BERYLLIUM		5. Report Date July 1970
		6. Performing Organization Code
7. Author(s) Lawrence E. Peters, Jr.		8. Performing Organization Report No. E-5473
9. Performing Organization Name and Address Lewis Research Center, National Aeronautics and Space Administration, Cleveland, Ohio 44135		10. Work Unit No. 120-27
		11. Contract or Grant No.
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration, Washington, D.C. 20546		13. Type of Report and Period Covered Technical Memorandum
		14. Sponsoring Agency Code
15. Supplementary Notes		

16. Abstract

The differential equations which describe the buildup of thermal neutron absorbers in beryllium have been solved. The solutions were used to calculate the change in the macroscopic absorption cross section of the Plum Brook Reactor as a result of the buildup of these poisons. This calculated value agrees within the experimental error of an actual observed change. The effects of variations in the factors on which the poison is dependent were studied and are presented in the form of graphs.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Irradiated beryllium Neutron absorber Neutron poison	Unclassified - unlimited

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price*
Unclassified	Unclassified	14	$3.00

*For sale by the Clearinghouse for Federal Scientific and Technical Information
Springfield, Virginia 22151

THERMAL NEUTRON ABSORBER BUILDUP IN IRRADIATED BERYLLIUM

by Lawrence E. Peters, Jr.

Lewis Research Center

SUMMARY

The differential equations which describe the buildup of thermal neutron absorbers in beryllium have been solved. The solutions were used to calculate the change in the macroscopic absorption cross section of the Plum Brook Reactor as a result of the buildup of these poisons. This calculated value agrees within the experimental error of an actual observed change. The effects of variations in the factors on which the poison is dependent were studied and are presented in the form of graphs.

INTRODUCTION

When beryllium is exposed to fast neutrons, a chain of reactions takes place which results in the buildup of isotopes with high thermal neutron cross sections. For thermal reactors using beryllium as a reflector or moderator material these thermal neutron poisons can be significant, especially if the reactor must be shut down for several months or more. For example, after being shut down for approximately 100 days the PBR (Plum Brook Reactor) critical rod bank height at restart was about 0.8 inch higher than normal.

The buildup of poison in beryllium has been analyzed before by Morris (ref. 1) for continuous reactor operation. However, his solutions for the equations governing the poison buildup are not general because of the use of boundary conditions for continuous operation.

In this work the differential equations governing the reactions were solved for more general boundary conditions. The change in the core absorption cross section was then calculated for cyclic reactor operation in the PBR. By using perturbation theory the corresponding change in reactivity was calculated. The analytical model was also used to study the effects of operation time, shutdown time, power, and fast to thermal flux ratio on the buildup of thermal neutron poisons in beryllium.

SYMBOLS

$\Delta K/K$	reactivity
N_1	number of Be^9 nuclei per cm^3
N_2	number of Li^6 nuclei per cm^3
N_3	number of H^3 nuclei per cm^3
N_4	number of He^3 nuclei per cm^3
N_i^0	value of Ni at beginning of time interval
$\delta\Sigma_a$	change in macroscopic absorption cross section
λ	decay constant for H^3, sec^{-1}
σ_1	cross section for Be^9 (n_f, α)
σ_2	cross section for Li^6 (n_s, α)
σ_4	cross section for He^3 (n_s, p)
φ_f	fast neutron flux $>$ 0.82 MeV, $n/(cm^2)(sec)$
φ_s	thermal neutron flux, $n/(cm^2)(sec)$
φ_s^*	adjoint thermal neutron flux

ANALYSIS

Defining Equations

The chain of reactions occurring in beryllium when in the presence of fast and thermal neutrons is as follows:

$$Be^9(n_f, \alpha) \rightarrow He^6(\beta^-)^{0.84S} \rightarrow Li^6(n_s, \alpha)^{910b} \rightarrow H^3 \underset{(n_s, p)5400b}{\overset{(\beta^-)12.5Y}{\rightleftarrows}} He^3$$

Two of the isotopes formed, Li^6 and He^3, have large thermal neutron absorption cross sections. To determine the concentration of these poisons, it was necessary to solve the set of simultaneous differential equations defining the reaction chain. These equations are

$$\frac{dN_2}{dt} = N_1 \varphi_f \sigma_1 - N_2 \varphi_s \sigma_2 \tag{1}$$

$$\frac{dN_3}{dt} = N_2 \varphi_s \sigma_2 - N_3 \lambda + N_4 \varphi_s \sigma_4 \tag{2}$$

$$\frac{dN_4}{dt} = N_3 \lambda - N_4 \varphi_s \sigma_4 \tag{3}$$

The above equations do not consider loss of hydrogen or helium by diffusion; however, for thick sections of beryllium (~1 in.) this is a negligible effect. The N_1 burnup is nearly zero, so for this analysis N_1 was assumed constant. The above equations were solved by using the Laplace transform and the complex inversion integral. The solutions give N_2, N_3, and N_4 as functions of time:

$$N_2 = \frac{N_1^0 \varphi_f \sigma_1}{\varphi_s \sigma_2}\left(1 - e^{-\varphi_s \sigma_2 t}\right) + N_2^0 e^{-\varphi_s \sigma_2 t} \tag{4}$$

$$\begin{aligned}
N_3 = {} & e^{-(\lambda + \varphi_s \sigma_4)t}\left\{N_3^0 - \frac{(N_3^0 + N_4^0)\varphi_s \sigma_4}{\lambda + \varphi_s \sigma_4} + \frac{(N_1^0 \varphi_f \sigma_1 + N_2^0 \varphi_s \sigma_4)\varphi_s \sigma_2}{[\lambda + \varphi_s(\sigma_4 - \sigma_2)](\lambda + \varphi_s \sigma_4)}\right. \\
& \left. - \frac{N_2^0 \varphi_s \sigma_2}{\lambda + \varphi_s(\sigma_4 - \sigma_2)} - \frac{N_1^0 \varphi_f \sigma_1 \varphi_s \sigma_2 \varphi_s \sigma_4}{(\lambda + \varphi_s \sigma_4)^2 [\lambda + \varphi_s(\sigma_4 - \sigma_2)]}\right\} \\
& + e^{-\varphi_s \sigma_2 t}\left\{\frac{N_1^0 \varphi_f \sigma_1 \varphi_s \sigma_4}{\varphi_s \sigma_2 [\lambda + \varphi_s(\sigma_4 - \sigma_2)]} - \frac{N_1^0 \varphi_f \sigma_1 + N_2^0 \varphi_s(\sigma_4 - \sigma_2)}{\lambda + \varphi_s(\sigma_4 - \sigma_2)}\right\} + \frac{N_1^0 \varphi_f \sigma_1 \varphi_s \sigma_4}{\lambda + \varphi_s \sigma_4} t \\
& + \frac{N_1^0 \varphi_f \sigma_1 + (N_2^0 + N_3^0 + N_4^0)\varphi_s \sigma_4}{\lambda + \varphi_s \sigma_4} - N_1^0 \varphi_f \sigma_1 \varphi_s \sigma_4 \left[\frac{1}{\varphi_s \sigma_s (\lambda + \varphi_s \sigma_4)} + \frac{1}{\lambda + \varphi_s \sigma_4}\right]
\end{aligned} \tag{5}$$

$$N_4 = e^{-(\lambda+\varphi_s\sigma_4)t}\left\{\frac{N_4^0\varphi_s\sigma_4}{\lambda + \varphi_s\sigma_4} + \frac{\left(N_2^0 + N_3^0\right)\varphi_s\sigma_2\lambda}{(\lambda + \varphi_s\sigma_4)[\lambda + \varphi_s(\sigma_4 - \sigma_2)]} - \frac{N_3^0\lambda}{\lambda + \varphi_s(\sigma_4 - \sigma_2)} - \frac{N_1^0\varphi_f\sigma_1\varphi_s\sigma_2\lambda}{(\lambda + \varphi_s\sigma_4)^2[\lambda + \varphi_s(\sigma_4 - \sigma_2)]}\right\}$$

$$+ e^{-\varphi_s\sigma_2 t}\left\{\frac{N_1^0\varphi_f\sigma_1}{\varphi_s\sigma_2} - N_2^0\right\}\frac{\lambda}{\lambda + \varphi_s(\sigma_4 - \sigma_2)} + \frac{N_1^0\varphi_f\sigma_1\lambda t}{\lambda + \varphi_s\sigma_4}$$

$$+ \left\{N_4^0 + N_3^0 + N_2^0 - N_1^0\varphi_f\sigma_1\left[\frac{1}{\varphi_s\sigma_2} + \frac{1}{\lambda + \varphi_s\sigma_4}\right]\right\}\frac{\lambda}{\lambda + \varphi_s\sigma_4} \tag{6}$$

With N_2 and N_4 determined, the increase in the macroscopic absorption cross section in the beryllium is given by

$$\delta\Sigma_a = N_2\sigma_2 + N_4\sigma_4 \tag{7}$$

A computer program was written to compute the poison buildup for cyclic reactor operation. (A Fortran listing of the code is given in the appendix of this report.) For cyclic operation the values of N_2, N_3, and N_4 at the end of an "on" or "off" portion of the cycle become the initial conditions for the next "off" or "on" portion of the cycle. During the reactor off time the values of φ_s and φ_f are negligible and taken as zero.

To calculate the reactivity change corresponding to a given change in macroscopic cross sections, perturbation theory was used. (See for example, ref. 2.) The reactivity change is given by

$$\Delta K/K = \frac{\delta\Sigma_a \int \varphi_s^*\varphi_s\, dv}{\nu \int \varphi_f^*\Sigma_f\varphi_s\, dv} \tag{8}$$

The adjoint flux integral was obtained using the Twenty-Grand diffusion theory program (ref. 3).

Application to Plum Brook Reactor

To determine the poison buildup in the PBR the beryllium reflector which surrounds the core on all four sides was broken up into seven regions. This was to account for the variation in flux around the reflector. Average fluxes for the seven regions were obtained from a two-dimensional neutron diffusion calculation. The value of σ_1 was determined by flux-weighting the cross section over the interval 0.82 MeV to 10 MeV. An average operation time and shutdown time per cycle was determined for the 5 years of operation prior to the 100 day shutdown. This information was input to the poison buildup computer program to solve for the poison atom densities. The reactivity corresponding to the change in absorption cross section was obtained from equation (8). The adjoint fluxes required to solve equation (8) were obtained from the neutron diffusion calculation. Table I gives some of the physical constants used for the PBR.

TABLE I. - VALUES OF CONSTANTS USED

FOR PBR CALCULATION

N_1^0, atoms/cm^3	1.12×10^{23}
N_2^0, atoms/cm^3	0
N_3^0, atoms/cm^3	0
N_4^0, atoms/cm^3	0
σ_1, b	0.0214
σ_2, b	950
σ_4, b	5400
λ, sec^{-1}	1.76×10^{-9}
φ_s, N/(cm^2)(sec)	$\sim5\times10^{14}$
φ_f (>0.82 MeV), N/(cm^2)(sec)	$\sim1\times10^{14}$
t_{on}, days	9.45
t_{off}, days	14.95

RESULTS AND DISCUSSION

Comparison of Measured and Calculated Reactivity Change

When the PBR was restarted following a 100-day shutdown, the critical rod bank height (RBH) was about 0.8 inch higher than expected. In terms of reactivity this is about \$1.35. The reactivity associated with the buildup of poisons in the beryllium reflector was calculated to be worth \$1.21. The estimated probable error in the predicted or expected reactivity (RBH), and hence in the measured reactivity change, is ±30 percent. The estimated probable error in the calculated reactivity change is believed to be about 20 percent.

Effects of Variations in the Independent Variables

Since there was a noticeable effect as a result of a 100-day shutdown, the question arose as to how serious the effect might be for longer shutdown periods. Figures 1 and 2 show the poison buildup as a function of shutdown time following 5 years of cyclic operation. For this period, which was the first 5 years of operation for the PBR, the average on time was about 10 days and the average down time was about 15 days. (Since that time this average has changed to about 20 days on and 5 days off.)



Figure 1. - Poison buildup for first year of shutdown time for Plum Brook Reactor after 5 years of cyclic operation.



Figure 2. - Poison buildup as function of shutdown time for Plum Brook Reactor after 5 years of cyclic operation.



Figure 3. - Poison buildup in Plum Brook Reactor beryllium for 20-day cycles.

The present PBR core and fuel loading are such that the maximum excess reactivity in the control rods is about $15.00. Thus from figure 2 it is seen that had the PBR been shut down for a 3 year period, it could not have been restarted without major core changes.

Figure 3 shows the effect of operation time for 20-day cycles. The longer the on time per cycle, the more quickly the poison level rises. However, after about 1 year of operation the poison level for 5 days operation out of 20 and 15 days operation out of 20 is about the same.

Figure 4 shows the poison buildup against operation time for several flux or power levels. (Figs. 4 to 6 are for continuous operation; however, the effects would be similar for cyclic operation.) The higher the reactor power, the steeper the rate of poison



Figure 4. - Poison buildup as function of operation time for neutron flux ratio $\varphi_f/\varphi_s = 1$.



Figure 5. - Poison buildup as function of operation time with fast neutron flux φ_f constant and thermal neutron flux φ_s a parameter.



Figure 6. - Poison buildup as function of operation time with thermal neutron flux φ_s constant and fast neutron flux φ_f a parameter.

buildup. However, after several years the poison level is independent of the power. After 10 years of operation the poison levels for the three different powers are within 3 percent of each other.

To examine the effect of the fast-to-thermal flux ratio, several combinations of the fast and thermal flux were calculated. The results are plotted in figures 5 and 6. Figure 5 shows that, for a given fast flux, changing the thermal flux causes a change in the poison level and the rate of poison buildup. Figure 6 shows that, for a given thermal flux, increasing the fast flux by a factor of 10 causes an increase in poison by a factor of 10.

CONCLUDING REMARKS

Following a long shutdown of the PBR there was an unexpected change in the core reactivity. This change can be accounted for by the buildup of Li^6 and He^3 in the beryllium reflector.

Although good agreement was obtained between the nominal measured and calculated reactivity change for the PBR, there was considerable uncertainty in both the measured value (because of uncertainty in the predicted critical rod position) and the calculated value. Chemical analyses are planned which will determine the atom densities of the reaction products in beryllium from the PBR core. This will give a second and more direct experimental check on the calculation of poison buildup in beryllium.

Lewis Research Center,
National Aeronautics and Space Administration,
Cleveland, Ohio, April 20, 1970,
120-27.

APPENDIX - IBM-1620 COMPUTER CODE FOR POISON BUILDUP IN BERYLLIUM FOR CYCLIC REACTOR OPERATION

```
C       POISON BUILDUP IN BERYLLIUM FOR CYCLIC REACTOR OPERATION
      5 PRINT 100
    100 FORMAT(57H1POISON BUILDUP IN BERYLLIUM FOR CYCLIC REACTOR OPERATIO
       1N,/)
        READ 200,AN1,AN2,AN3,AN4,PHI1,PHI4,NTS,TON,TOFF
    200 FORMAT(4E9.4,2E9.2,I4,2F6.0)
        T=TON*86400.
        DK=1.76E-9
        SIG1=.0214E-24
        SIG2=950.E-24
        SIG4=5400.E-24
        PRINT 101,AN1
        PRINT 102,AN2
        PRINT 103,AN3
        PRINT 104,AN4
        PRINT 105,PHI1
        PRINT 106,PHI4
    101 FORMAT(23H INITIAL BE NO. DEN. IS,E10.2)
    102 FORMAT(23H INITIAL LI NO. DEN. IS,E10.2)
    103 FORMAT(23H INITIAL TR NO. DEN. IS,E10.2)
    104 FORMAT(23H INITIAL HE NO. DEN. IS,E10.2/)
    105 FORMAT(27H FAST (GRP. 1) FLUX USED IS,E10.2)
    106 FORMAT(27H SLOW (GRP. 4) FLUX USED IS,E10.2/   )
        PRINT 117,TON
        PRINT 118,TOFF
    117 FORMAT (30H IRRADIATION TIME PER CYCLE IS,F6.2,5H DAYS)
    118 FORMAT (27H SHUTDOWN TIME PER CYCLE IS,F9.2,5H DAYS//)
        PHIF=PHI1
        PHIS=PHI4
        I=0
        M=0
      1 I=I+1
        A=AN1*PHIF*SIG1
        B=PHIS*SIG2
        C=PHIS*SIG4
        D=DK+C
        E=D-B
        EB=EXP(-B*T)
        ED=EXP(-D*T)
        BN2=A/B*(1.-EB)+AN2*EB
        BN3=ED*(AN3-(AN3+AN4)*C/D+(A+AN2*C)*B/E/D-AN2*B/E-A*B*C/D/D/E)+EB*
       1(AN2*B/E-(A+AN2*C)/E+A*C/B/E)+A*C/D*T+(AN3+AN4)*C/D+(A+AN2*C)/D-A*
       2C*(1./B/D+1./D/D)
        BN4=ED*(AN4*C/D+(AN2+AN3)*B*DK/D/E-AN3*DK/E-A*B*DK/D/D/E)+EB*(A/B-
       1AN2)*DK/E+A*DK/D*T+(AN4+AN3+AN2-A*(1./B+1./D))*DK/D
        AN2=BN2
```

```
      AN3=BN3
      AN4=BN4
      PN2=AN2*SIG2
      PN4=AN4*SIG4
      DELSIG=PN2+PN4
      PRINT 108,AN1
      PRINT 109,AN2
      PRINT 110,AN3
      PRINT 111,AN4
      PRINT 112,PN2
      PRINT 113,PN4
      PRINT 114,DELSIG
      PRINT 115,I
  108 FORMAT (18H BE NO. DENSITY IS,E10.2)
  109 FORMAT (18H LI NO. DENSITY IS,E10.2)
  110 FORMAT (18H TR NO. DENSITY IS,E10.2)
  111 FORMAT (18H HE NO. DENSITY IS,E10.2/)
  112 FORMAT (25H DELTA SIGMA DUE TO LI IS,E10.2)
  113 FORMAT (25H DELTA SIGMA DUE TO HE IS,E10.2)
  114 FORMAT (25H THE TOTAL DELTA SIGMA IS,E10.2/)
  115 FORMAT (44H THE ABOVE VALUES ARE AFTER TIME STEP NUMBER,I4,//)
      IF(M)2,2,3
    2 M=1
      PHIF=1.E-20
      PHIS=1.E-20
      T=TOFF*86400.
      IF(I-NTS)1,4,4
    3 M=0
      PHIF=PHI1
      PHIS=PHI4
      T=TON*86400.
      IF(I-NTS)1,4,4
    4 PRINT 116
  116 FORMAT (19H END OF CALCULATION)
      GO TO 5
      END
```

REFERENCES

1. Morris, W. L.: Calculation of Rate of Poison Buildup in Beryllium Reflector in MTR. Rep. IDO-16024, Phillips Petroleum Co., July 1, 1952.

2. Meghreblian, Robert V.; and Holmes, David K.: Reactor Analysis. McGraw-Hill Book Co., Inc., 1960, ch. 13.

3. Tobias, M. L.; and Fowler, T. B.: The Twenty Grand Program for the Numerical Solution of Few-Group Neutron Diffusion Equations in Two Dimensions. Rep. ORNL-3200, Oak Ridge National Lab., Feb. 21, 1962.

"The aeronautical and space activities of the United States shall be conducted so as to contribute . . . to the expansion of human knowledge of phenomena in the atmosphere and space. The Administration shall provide for the widest practicable and appropriate dissemination of information concerning its activities and the results thereof."

— NATIONAL AERONAUTICS AND SPACE ACT OF 1958

NASA SCIENTIFIC AND TECHNICAL PUBLICATIONS

TECHNICAL REPORTS: Scientific and technical information considered important, complete, and a lasting contribution to existing knowledge.

TECHNICAL NOTES: Information less broad in scope but nevertheless of importance as a contribution to existing knowledge.

TECHNICAL MEMORANDUMS: Information receiving limited distribution because of preliminary data, security classification, or other reasons.

CONTRACTOR REPORTS: Scientific and technical information generated under a NASA contract or grant and considered an important contribution to existing knowledge.

TECHNICAL TRANSLATIONS: Information published in a foreign language considered to merit NASA distribution in English.

SPECIAL PUBLICATIONS: Information derived from or of value to NASA activities. Publications include conference proceedings, monographs, data compilations, handbooks, sourcebooks, and special bibliographies.

TECHNOLOGY UTILIZATION PUBLICATIONS: Information on technology used by NASA that may be of particular interest in commercial and other non-aerospace applications. Publications include Tech Briefs, Technology Utilization Reports and Notes, and Technology Surveys.

Details on the availability of these publications may be obtained from:

SCIENTIFIC AND TECHNICAL INFORMATION DIVISION

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

Washington, D.C. 20546